

# A Sheet of Maple to Compute Second-Order Edgeworth Expansions and Related Quantities of any Function of the Mean of an iid Sample of an Absolutely Continuous Distribution

F. Bertrand<sup>a\*\*</sup> and M. Maumy-Bertrand<sup>a</sup>  
<sup>a</sup>*Institut de Recherche Mathématique Avancée,*

*Université de Strasbourg,  
 7, rue René Descartes, 67084 Strasbourg Cedex, France*

January 30, 2022

## Abstract

We designed a completely automated **Maple** ( $\geq 15$ ) worksheet for deriving Edgeworth and Cornish-Fisher expansions as well as the acceleration constant of the bootstrap bias-corrected and accelerated technique. It is valid for non-parametric or parametric bootstrap, of any (studentized) statistics that is -a regular enough- function of the mean of an iid sample of an absolutely continuous distribution.

This worksheet allowed us to point out one error in the second-order Cornish-Fisher expansion of the studentized mean stated in Theorem 13.5 by Das Gupta in [Das Gupta, 2008, p. 194] as well as lay the stress on the influence of the slight change of the normalizing constant when computing the second-order Edgeworth and Cornish-Fisher expansions of the  $t$ -distribution as stated in Theorem 11.4.2 by Lehman and Romano in [Lehman and Romano, 2005, p. 460].

In addition, we successfully applied the worksheet to a complex maximum likelihood estimator as a first step to derive more accurate confidence intervals in order to enhance quality controls.

The worksheet also features export of **Maple** results into **R** code. In addition, we provide **R** code to plot these expansions as well as their increasing rearrangements. All these supplemental materials are available upon request.

## 1 Introduction

Even though the Edgeworth expansion is often, at the first glance, mostly viewed as a theoretical tool, it has steadily gained interest from the practitioner point of view for several decades. At the time present, there are too many examples of smart applications of these expansions to list them all. For instance, it has successfully been used to improve confidence intervals, either asymptotic or bootstrap ones, giving sharp insights on, among other properties, their coverage or expected length. Hall [1992] and Shao and Tu [1996] provide a detailed account on this

---

<sup>\*\*</sup>Corresponding author. Email: Frederic.Bertrand@math.unistra.fr

topic. Moreover, the Edgeworth expansion is deeply rooted in one of the more recent outstanding study dealing with confidence interval for a binomial proportion (Brown, Cai, and Das Gupta, Brown et al. [2001, 2002]), in an exponential family (Brown, Cai, and Das Gupta, Brown et al. [2003]) and in discrete distributions (Cai, Cai [2005]).

The major drawback of these expansions are their long and tedious computations. Even when dealing with the most simple statistic -the studentized mean-, a famous textbook can provide an incorrect formula as the second-order Cornish-Fisher expansion stated in Theorem 13.5 by Das Gupta in [Das Gupta, 2008, p. 194]. Make a slight change in the denominator of the usual raw variance estimator to use instead its unbiased counterpart -by simply replacing  $1/\sqrt{n}$  by  $1/\sqrt{n-1}$  to get an exactly  $t$ -distributed statistic in the Gaussian framework- and it leads to different second-order formulas, which accounts for the discrepancies of the Edgeworth expansions for the studentized mean when compared to the Edgeworth expansion of the  $t$ -distribution, for instance as stated in Theorem 11.4.2 by Lehman and Romano in [Lehman and Romano, 2005, p. 460].

For a smooth -or regular enough- model  $A(\bar{X})$  of the mean  $\bar{X}$  of an iid sample  $(X_1, \dots, X_n)$  drawn from an absolutely continuous distribution  $\mathcal{L}_\theta$ , first and second-order Edgeworth expansions are valid. Closed, yet complex, formulas were derived by Withers [1983] and Hall [1992] to compute them from the partial derivatives of  $A$  and the central moments of the underlying distribution  $\mathcal{L}_\theta$ . In addition, in such a model that exhibits a valid second-order Edgeworth expansion, there exists a closed formula for the acceleration constant  $a$  used in the bootstrap bias-corrected and accelerated technique (BCA). Were we able to evaluate this formula, this would help to remedy to “A disadvantage of the bootstrap BCA”, as stated Shao and Tu [Shao and Tu, 1996, p. 140], which is “that the determination of  $a$  is not easy, especially when the problem under consideration is complex”. Moreover, in some statistical models, as put forward in two cases in Sections 4 and 5, the acceleration constant  $a$  is free of the unknown parameters of the underlying distribution and thus can be derived exactly.

All these reasons account for our use of the **Maple** computation engine Maple Team [2014], that combines numeric computations with symbolic capabilities, in order to design a tool able to automatically and trustworthily compute Edgeworth expansions as well as Cornish-Fisher ones or the acceleration constant  $a$  used in BCA bootstrap technique. This is this tool, that we provide as a **Maple** worksheet.

**Maple** functions or expressions code can be translated into **MATLAB** code, which in our setting, is virtually **R** Core Team [2014] code. The **R** language can then be used for display purposes, as we did with Figures 1 and 2, to apply bootstrap techniques of the **boot** package Canty and Ripley [2014], Davison and Hinkley [1997] or for any further statistical analysis, for instance to deal with the non-increasing cumulative distribution functions that may result from the above Edgeworth expansions using the rearrangement operator Hardy et al. [1952] implemented in the **Rearrangement** package Chernozhukov et al. [2010], Graybill et al. [2011]. Using Withers [1983], it is easy to derive Cornish-Fisher expansions of any higher order and upgrade our worksheet to have **Maple** compute them.

Previous work by Finner and Dickhaus Finner and Dickhaus [2010] only addresses the part on the computation of Edgeworth expansions up to any order. From a (applied) statistical point of view, it is not very useful. On the contrary, we not only carry out the Edgeworth expansions but also the Cornish-Fisher ones, BCA acceleration constant derivation, export to and import in **R** of the results, then use of the **R** language to derive increasing rearrangements of the computed expansions, which sums up to a significant additional value of this article for statisticians and practitioners.

## 2 Edgeworth-Cornish-Fisher expansions and BCA bootstrap

### 2.1 Introducing the Edgeworth expansion

Let  $\{\mathbf{Z}_n : n \geq 1\}$  be a sequence of random variables with 0 mean and unit variance. The cumulant generating function  $\Psi_n$  of  $\mathbf{Z}_n$  is defined by:  $\Psi_n(t) = \log \mathbb{E}[it\mathbf{Z}_n]$ ,  $t \in \mathbb{R}$ . The  $\nu^{\text{th}}$  cumulant of  $\mathbf{Z}_n$  is given by:  $\kappa_{\nu,n} = \frac{1}{i^\nu} \frac{d^\nu}{dt^\nu} \Psi(0)$ ,  $\nu \in \mathbb{N}$ . Consequently, we show that:

$$\mathbb{E}[\exp(it\mathbf{Z}_n)] = \exp\left(\sum_{j=1}^{+\infty} \kappa_{j,n} \frac{1}{j!} (it)^j\right), \quad t \in \mathbb{R}. \quad (1)$$

The  $\nu^{\text{th}}$  derivative of the characteristic function of  $\mathbf{Z}_n$  exists at the point  $t = 0$  if and only if  $\mathbb{E}[|\mathbf{Z}_n|^\nu] < +\infty$ . Then the  $\nu^{\text{th}}$  cumulant of  $\mathbf{Z}_n$  exists if and only if  $\mathbb{E}[|\mathbf{Z}_n|^\nu] < +\infty$ . The entire series expansion and the identification of the coefficients lead to the expression of the first four cumulants, see Hall [1992] for more details:

$$\begin{aligned} \kappa_{1,n} &= \mathbb{E}[\mathbf{Z}_n], & \kappa_{2,n} &= \text{Var}[\mathbf{Z}_n], \\ \kappa_{3,n} &= \mathbb{E}[\mathbf{Z}_n - \mathbb{E}[\mathbf{Z}_n]]^3, & \kappa_{4,n} &= \mathbb{E}[\mathbf{Z}_n - \mathbb{E}[\mathbf{Z}_n]]^4 - 3(\text{Var}[\mathbf{Z}_n])^2. \end{aligned}$$

Assume that there exist four real numbers  $k_{1,2}$ ,  $k_{2,2}$ ,  $k_{3,1}$ ,  $k_{4,1}$  so that the first four cumulants satisfy the following asymptotic expansions:

$$\begin{aligned} \kappa_{1,n} &= n^{-1/2}k_{1,2} + O(n^{-3/2}), & \kappa_{2,n} &= 1 + n^{-1}k_{2,2} + O(n^{-3/2}), \\ \kappa_{3,n} &= n^{-1/2}k_{3,1} + O(n^{-3/2}), & \kappa_{4,n} &= n^{-1}k_{4,1} + O(n^{-3/2}). \end{aligned}$$

Then Equation 1 becomes:

$$\begin{aligned} \mathbb{E}[\exp(it\mathbf{Z}_n)] &= \exp\left(-\frac{1}{2}t^2\right) \left\{ 1 + n^{-1/2} \left( k_{1,2}it + \frac{1}{6}k_{3,1}(it)^3 \right) \right. \\ &\quad \left. + n^{-1} \left( \frac{1}{2}k_{2,2}(it)^2 + \frac{1}{24}k_{4,1}(it)^4 + \frac{1}{2} \left( k_{1,2}it + \frac{1}{6}k_{3,1}(it)^3 \right)^2 \right) + \dots \right\}. \end{aligned}$$

A Fourier inversion of this expansion leads to:

$$\mathbb{P}[\mathbf{Z}_n \leq x] = \Phi(x) + \left( n^{-1/2}p_1(x) + n^{-1}p_2(x) + \dots \right) \phi(x),$$

uniformly in  $x \in \mathbb{R}$  where  $\phi(x) = \Phi'(x)$  is the probability density function of a standard Gaussian distribution and

$$p_1(x) = - \left( k_{1,2} + \frac{k_{3,1}}{6}(x^2 - 1) \right), \quad (2)$$

$$p_2(x) = -x \left( \frac{(k_{2,2} + k_{1,2}^2)}{2} + \frac{(k_{4,1} + 4k_{1,2}k_{3,1})}{24}(x^2 - 3) + \frac{k_{3,1}^2}{72}(x^4 - 10x^2 + 15) \right). \quad (3)$$

## 2.2 A model valid for Edgeworth expansions

Let  $\mathbf{X}, \mathbf{X}_1, \mathbf{X}_2, \dots$  be independent and identically distributed random column  $d$ -vectors with mean  $\mathbb{E}[\mathbf{X}]$  denoted by  $\boldsymbol{\mu}$ , and  $\bar{\mathbf{X}} = n^{-1} \sum_{i=1}^n \mathbf{X}_i$ . Let  $A : \mathbb{R}^d \rightarrow \mathbb{R}$  be a smooth function satisfying  $A(\boldsymbol{\mu}) = 0$ . Denote the  $i$ th element of a  $d$ -vector  $\mathbf{v}$  by:  $v^{(i)}$  or  $(\mathbf{v})^{(i)}$ ,

$$a_{i_1 \dots i_j} = (\partial^j / \partial x^{(i_1)} \dots \partial x^{(i_j)}) A(\mathbf{X})|_{\mathbf{X}=\boldsymbol{\mu}} \quad (4)$$

and

$$\mu_{i_1 \dots i_j} = \mathbb{E} \left[ (\mathbf{X} - \boldsymbol{\mu})^{(i_1)} \dots (\mathbf{X} - \boldsymbol{\mu})^{(i_j)} \right]. \quad (5)$$

We will focus on functions such as:

1.  $A(\mathbf{x}) = (g(\mathbf{x}) - g(\boldsymbol{\mu}))/h(\boldsymbol{\mu})$ , where  $\theta_0 = g(\boldsymbol{\mu})$  is the (scalar) parameter estimated by  $\hat{\theta} = g(\bar{\mathbf{X}})$  and  $h(\boldsymbol{\mu})^2$  is the asymptotic variance  $\sigma_{\hat{\theta}}^2$  of  $n^{1/2}\hat{\theta}$ ;
2.  $A(\mathbf{x}) = (g(\mathbf{x}) - g(\boldsymbol{\mu}))/h(\mathbf{x})$ , where  $h(\bar{\mathbf{X}})$  is an estimate of  $h(\boldsymbol{\mu}) = \sigma_{\hat{\theta}}$ .

We can assume that  $h$  is a known function; as stated in [Hall, 1992], the following  $h$  function fulfills the previous requirements:  $h^2(\mathbf{x}) = \sum_{i=1}^d \sum_{j=1}^d (\partial g / \partial x^{(i)} \partial g / \partial x^{(j)}) (\mathbf{x}) \mu_{i,j}$ . If used,  $A_0$  stands for the non-studentized case and  $A_s$  the studentized one. This “smooth function model” [Hall, 1992] allows to study problems where  $\theta_0$  is a mean, or a variance, or a ratio of means or variances, or a difference of means or variances, or a correlation coefficient, etc.

*Example 2.1.* Let  $\{W_1, \dots, W_n\}$  be a random sample from an univariate population with mean  $m$  and variance  $\beta^2$ , and if we wished to estimate  $\theta_0 = m$ , then we would take  $d = 2$ ,  $\mathbf{X} = (X^{(1)}, X^{(2)})^\top = (W, W^2)^\top$ ,  $\boldsymbol{\mu} = \mathbb{E}[\mathbf{X}]$ ,  $g(x^{(1)}, x^{(2)}) = x^{(1)}$ ,  $h^2(x^{(1)}, x^{(2)}) = x^{(2)} - (x^{(1)})^2$ . This would ensure that  $g(\boldsymbol{\mu}) = m$ ,  $g(\bar{\mathbf{X}}) = \bar{W}$  (the sample mean),  $h(\boldsymbol{\mu}) = \beta^2$ , and

$$h^2(\bar{\mathbf{X}}) = n^{-1} \sum_{i=1}^n X_i^{(2)} - \left( n^{-1} \sum_{i=1}^n X_i^{(1)} \right)^2 = n^{-1} \sum_{i=1}^n (W_i - \bar{W})^2 = \hat{\beta}^2 \text{ (the sample variance).}$$

*Example 2.2.* Let  $\{W_1, \dots, W_n\}$  be a random sample from an univariate population with mean  $m$  and variance  $\beta^2$ , and if we wished to estimate  $\theta_0 = \beta^2$ , then we would take  $d = 4$ ,  $\mathbf{X} = (W, W^2, W^3, W^4)^\top$ ,  $\boldsymbol{\mu} = \mathbb{E}[\mathbf{X}]$ ,  $g(x^{(1)}, x^{(2)}, x^{(3)}, x^{(4)}) = x^{(2)} - (x^{(1)})^2$ ,  $h(x^{(1)}, x^{(2)}, x^{(3)}, x^{(4)}) = x^{(4)} - 4x^{(1)}x^{(3)} + 6(x^{(1)})^2x^{(2)} - 3(x^{(1)})^4 - (x^{(2)} - (x^{(1)})^2)^2$ . In this case, we have:

$$g(\boldsymbol{\mu}) = \beta^2, \quad g(\bar{\mathbf{X}}) = \hat{\beta}^2, \quad h(\boldsymbol{\mu}) = \mathbb{E}[W - m]^4 - \beta^4, \quad h(\bar{\mathbf{X}}) = n^{-1} \sum_{i=1}^n (W_i - \bar{W})^4 - \hat{\beta}^4.$$

The cases where  $\theta_0$  is a correlation coefficient (a function of five means), or a variance ratio (a function of four means), among others, may be treated similarly.

Since  $A(\boldsymbol{\mu}) = 0$  and  $n^{1/2}(\bar{\mathbf{X}} - \boldsymbol{\mu}) = O_{\mathbb{P}}(1)$ , by a Taylor expansion:  $S_n = n^{1/2}A(\bar{\mathbf{X}}) = S_{nr} + O_{\mathbb{P}}(n^{-r/2})$ ,  $r \geq 1$ , where  $S_n - S_{nr} = O_{\mathbb{P}}(n^{-r/2})$  means that  $\lim_{\lambda \rightarrow +\infty} \limsup_{n \rightarrow +\infty} \mathbb{P}(|S_n - S_{nr}| / n^{-r/2} > \lambda) = 0$  and

$$S_{nr} = \sum_{i=1}^d a_i Z^{(i)} + \frac{n^{-1/2}}{2} \sum_{i_1=1}^d \sum_{i_2=1}^d a_{i_1 i_2} Z^{(i_1)} Z^{(i_2)} + \dots + \frac{n^{-(r-1)/2}}{r!} \sum_{i_1=1}^d \dots \sum_{i_r=1}^d a_{i_1 \dots i_r} Z^{(i_1)} \dots Z^{(i_r)}.$$

**Theorem 2.3** (Hall Hall [1992], Bertrand Bertrand and Maumy [2007]). *Assume that the function  $A$  has 4 continuous derivatives in a neighborhood of  $\boldsymbol{\mu} = \mathbb{E}[\mathbf{X}]$ , that  $A(\boldsymbol{\mu}) = 0$ , that*

$\mathbb{E}[\|\mathbf{X}\|^4] < +\infty$ , and that the characteristic function  $\chi$  of  $\mathbf{X}$  satisfies  $\limsup_{\|\mathbf{t}\| \rightarrow +\infty} |\chi(\mathbf{t})| < 1$ . We have:

$$\mathbb{P}[\sqrt{n}A(\bar{\mathbf{X}}) \leq x] = \Phi(x) + n^{-1/2}p_1(x)\phi(x) + n^{-1}p_2(x)\phi(x) + O\left(n^{-3/2}\right), \quad (6)$$

uniformly in  $x \in \mathbb{R}$  and where  $p_1, p_2$  are defined by Equations 2 and 3, the values of the coefficients  $k_{1,2}, k_{2,2}, k_{3,1}$  et  $k_{4,1}$  been given by:

$$\begin{aligned} k_{1,2} &= \frac{1}{2} \sum_{i=1}^d \sum_{j=1}^d a_{ij} \mu_{ij}, \\ k_{2,2} &= \sum_{i=1}^d \sum_{j=1}^d \sum_{k=1}^d a_i a_{jk} \mu_{ijk} + \frac{1}{2} \sum_{i=1}^d \sum_{j=1}^d \sum_{k=1}^d \sum_{l=1}^d a_{ij} a_{kl} \mu_{ik} \mu_{jl} \\ &\quad + \sum_{i=1}^d \sum_{j=1}^d \sum_{k=1}^d \sum_{l=1}^d a_i a_{jkl} \mu_{ij} \mu_{kl}, \\ k_{3,1} &= \sum_{i=1}^d \sum_{j=1}^d \sum_{k=1}^d a_i a_j a_k \mu_{ijk} + 3 \sum_{i=1}^d \sum_{j=1}^d \sum_{k=1}^d \sum_{l=1}^d a_i a_j a_{kl} \mu_{ik} \mu_{jl}, \\ k_{4,1} &= \sum_{i=1}^d \sum_{j=1}^d \sum_{k=1}^d \sum_{l=1}^d a_i a_j a_k a_l (\mu_{ijkl} - 3\mu_{ij} \mu_{kl}) \\ &\quad + 12 \sum_{i=1}^d \sum_{j=1}^d \sum_{k=1}^d \sum_{l=1}^d \sum_{m=1}^d a_i a_j a_k a_{lm} \mu_{il} \mu_{jkm} \\ &\quad + 12 \sum_{i=1}^d \sum_{j=1}^d \sum_{k=1}^d \sum_{l=1}^d \sum_{m=1}^d \sum_{o=1}^d a_i a_j a_{kl} a_{mo} \mu_{ik} \mu_{jm} \mu_{lo} \\ &\quad + 4 \sum_{i=1}^d \sum_{j=1}^d \sum_{k=1}^d \sum_{l=1}^d \sum_{m=1}^d \sum_{o=1}^d a_i a_j a_k a_{lmo} \mu_{il} \mu_{jm} \mu_{ko}. \end{aligned}$$

Using Hall [1992] and Withers [1983], it is easy to derive Edgeworth expansions of any higher order and complete our worksheet to have **Maple** compute them. One can also try to use Finner and Dickhaus work Finner and Dickhaus [2010].

*Remark 2.4.* Edgeworth expansions can also be used to derive asymptotic expansions for cumulative density functions of distribution. A case in point is the  $t$ -distribution, see Lehman and Romano in Formulas 11.73 and 11.74 of Theorem 11.4.2, [Lehman and Romano, 2005, p. 460] for the second-order Edgeworth expansion of the  $t$ -distribution. An easy way to retrieve an exactly  $t$  distributed statistic in a iid  $(X_1, \dots, X_n)$  Gaussian  $\mathcal{N}(\mu, \sigma)$  framework, is to consider  $\sqrt{n}(\bar{X} - \mu)/\hat{\sigma}_c$  with  $\hat{\sigma}_c$  the square root of the unbiased variance estimator  $(1/(n-1) \sum_{i=1}^n (X_i - \bar{X})^2)$ . Unfortunately, when applying Theorem 2.3 to the sample mean, we get the expansion of  $\sqrt{n}(\bar{X} - \mu)/\hat{\sigma}$  with  $\hat{\sigma}$  the square root of the raw variance estimator  $(1/n \sum_{i=1}^n (X_i - \bar{X})^2)$ . This subtle change in the normalizing constant leads to discrepancies between the two second-order terms of the Edgeworth expansion of the  $t$ -distribution and of the studentized mean. Using asymptotic expansions, we show in the provided Maple worksheet, and report in Section 3.4, how to switch from the expansion of the studentized mean to the one of the  $t$ -distribution.

## 2.3 Cornish-Fisher expansion

The Cornish-Fisher expansion is the expansion of the  $\alpha$ -level quantile of an estimator from its Edgeworth expansion:

$$w_\alpha = \inf \{x : \mathbb{P} [\sqrt{n}A(\bar{\mathbf{X}}) \leq x] \geq \alpha\}.$$

Denote by  $z_\alpha$  the  $\alpha$  level quantile of a standard Gaussian distribution. Equation 6 leads to an expansion of  $w_\alpha$  as a polynomial in  $z_\alpha$ :

$$w_\alpha = z_\alpha + n^{-1/2}p_{11}(z_\alpha) + n^{-1}p_{21}(z_\alpha) + O(n^{-3/2})$$

where  $p_{11}(x) = -p_1(x)$  and  $p_{21}(x) = p_1(x)p_1'(x) - \frac{1}{2}xp_1^2(x) - p_2(x)$  with  $p_1$  and  $p_2$  defined by Equations 2 and 3.

Even though the true values of the parameter  $\theta_0$ , whose value may be required to compute the polynomials  $p_1$  and  $p_2$  in the Cornish-Fisher expansion when working with non pivotal statistics, are unknown, Hall [1992] and Shao and Tu [1996] showed that this issue can be solved using bootstrap techniques to ensure the interval still features the second-order probability matching property.

The same remark as above applies, using Withers [1983], it is easy to derive Cornish-Fisher expansions of any higher order and complete our worksheet to have **Maple** compute them.

## 2.4 BCA bootstrap

In a framework valid for second-order Edgeworth expansions, bootstrap bias corrected and accelerated (BCA) are second-order probability matching intervals [Hall, 1992, Shao and Tu, 1996]. Yet, even though the simpler bootstrap  $t$  intervals also feature second-order probability matching, BCA ones are often preferred [Hall, 1992, p. 93 and pp. 133-134] and their definition will be now shortly recalled. Note that the BCA intervals are transformation respecting and thus invariant under re-parametrization.

Let  $\chi$  be a random sample from the population studied and  $\theta$  a parameter of interest. Define  $\hat{H}(x) = \mathbb{P}[\hat{\theta}^* \leq x | \chi]$  for the bootstrap distribution of  $\hat{\theta}^*$  and, for all  $\alpha$  in  $]0; 1[$ ,  $\hat{H}^{-1}(\alpha) = \inf\{x : \hat{H}(x) \geq \alpha\}$ . In this notation,  $\hat{y}_\alpha = \hat{H}^{-1}(\alpha) = \hat{H}^{-1}(\Phi(z_\alpha))$ . The sequences of quantiles  $\{\hat{y}_\alpha, 0 < \alpha < 1\}$  would be “centered” empirically if  $\hat{y}_{1/2}$  were equal to  $\hat{\theta}$ . In general, this identity does not hold, and we might wish to correct for centering error. Let  $\hat{m} = \Phi^{-1}\{\hat{H}(\hat{\theta})\}$  be the estimate of the centering correction or bias correction. An one-sided BCA interval for  $\theta_0$ , with nominal coverage  $1 - \alpha$ , is given by  $(-\infty, \hat{y}_{BCA, 1-\alpha})$  where:

$$\hat{y}_{BCA, 1-\alpha} = \hat{H}^{-1}(\Phi[\hat{m} + (\hat{m} + z_{1-\alpha})(1 - \hat{a}(\hat{m} + z_{1-\alpha}))^{-1}]).$$

The quantity  $(1 - \hat{a}(\hat{m} + z_{1-\alpha}))^{-1}$  amounts for a skewness correction. An explicit formula for the acceleration constant  $a$ , valid for non-parametric as well as parametric bootstrap, is worked out by Hall [Hall, 1992, p. 132]:

$$a = \frac{A}{6\sigma^3\sqrt{n}}, \text{ where } A = \sum_{i=1}^d \sum_{j=1}^d \sum_{k=1}^d a_i a_j a_k \mu_{ijk}.$$

An estimate is also proposed by Hall [Hall, 1992, p. 133]:

$$\hat{a} = \frac{\hat{A}}{6\hat{\sigma}^3\sqrt{n}}, \text{ where } \hat{A} = \sum_{i=1}^d \sum_{j=1}^d \sum_{k=1}^d \hat{a}_i \hat{a}_j \hat{a}_k \hat{\mu}_{ijk}$$

with  $\hat{a}_i = \left( \partial / \partial x^{(i)} \right) A(\mathbf{X})|_{\mathbf{X}=\boldsymbol{\mu}}$  and  $\hat{\mu}_{ijk} = \frac{1}{n} \sum_{l=1}^n \left[ (\mathbf{X}_l - \boldsymbol{\mu})^{(i)} (\mathbf{X}_l - \boldsymbol{\mu})^{(j)} (\mathbf{X}_l - \boldsymbol{\mu})^{(k)} \right]$ .

A two-sided BCA interval is given by  $(\hat{y}_{BCA, \alpha/2}, \hat{y}_{BCA, 1-\alpha/2})$ .

### 3 Expanding the mean

Let  $\Gamma_1$  be the skewness and  $\kappa_1$  the excess kurtosis. All the explicit results in this section were derived from the theoretical ones stated in Section 2 using **Maple**. For this first application we will follow closely the steps detailed in the file `SI_BertrandMaumy_Edgeworth_mean.mw`.

#### 3.1 Setting the framework

We begin with a restart statement to be sure to start with a fresh session.

```
> restart;
```

We first need to set within **Maple** the framework of the Exemple 2.1 in Section 2.2. We want to highlight that the user only needs to define the  $g$  function, see Section 2.2, all the other quantities required for the computation, for both the non-studentized or studentized cases, been automatically derived from this very definition as soon as the previous function remained called  $g$ .

```
> g:=proc(x1) return x1; end proc;
```

```
g := proc(x1) return x1 end proc;
```

Automatically retrieve the value of the dimension  $d$ , see Section 2.2, for the non-studentized case (`Dim:=1`) and for the studentized one (`Dims:=2`). Enlarge the vector to take into account higher moments required for the computations related to the studentized statistic.

```
> Dim:=nops([op(1,op(1,g))]); Dims:=2*Dim; NtoL:=[evaln(x||(1 .. Dims))];
```

```
Dim := 1
```

```
Dims := 2
```

```
NtoL := [x1, x2]
```

Even the  $h$  function is automatically defined by this code chunk.

```
> h := proc (x::(seq(name))) option remember; local NNN, derg;
for NNN in seq(1..(_npassed/2)) do
derg[NNN]:=diff(g(op([1..Dim],[_passed])),
[op(NNN,[_passed])]);
end do;
return sqrt(sum(sum(derg[iii]*derg[jjj]*(op(iii+jjj,[_passed])-op(iii,
[_passed])*op(jjj,[_passed])),jjj=1..(_npassed/2)),iii=1..(_npassed/2)));
end proc;
```

```

h := proc(x :: seq(name))
local NNN, derg;
option remember;
for NNN in seq(1..1/2 * nargs) do
  derg[NNN] := diff(g(op([1..Dim], [args])), [op(NNN, [args])])
end do;
return sqrt(sum(sum(derg[iii] * derg[jjj] * (op(iii + jjj, [args]) - op(iii, [args])
  * op(jjj, [args])), jjj = 1..1/2 * nargs), iii = 1..1/2 * nargs))
end proc;

```

We begin with the link between raw moments, from first to fourth order, and centered ones. We assume  $\mu_3 = \Gamma_1$  is the skewness,  $\mu_4 = \kappa_1 + 3$  the excess kurtosis and  $\mu_i = E((X - \mu)^i)/\sigma^i$ ,  $i \geq 5$ , standardized central moments. Can be tuned with a specific distribution. The variable names  $\Gamma_1$  and  $\kappa_1$  had to be used to avoid errors due to confusion with two built-in R functions `Gamma` and `kappa` if the result is exported to the R statistical language.

```

> EspX:=[]:for ii in [seq(1..nops(NtoL))]
do EspX:=op(EspX),cat(x,ii)=expand(sum(binomial(ii, j)*'mu||j'*
  sigma^j*mu^(ii-j), 'j' = 0 .. ii)): end do:
EspX;

```

$$[x1 = \mu, x2 = \mu^2 + \sigma^2]$$

We reflect some known facts about the raw moments so that Maple can use them when simplifying complex expressions.

```

> for AA in [seq(1..nops(NtoL))] do if 'mod'(AA,2)=1 then
  assume(NtoL[AA],real) else assume(NtoL[AA]>0) end if end do;
NtoL;

```

$$[x1, x2]$$

Several code chunks, not shown, now define various quantities of interests either for the computation itself or for speeding it up, namely:  $SigmaA = h(\mu)$ ,  $A0$ , its partial derivatives  $partA0$ ,  $As$ , its partial derivatives  $partAs$ , see Section 2.2. The last preparation step, is to compute all the moments that appear in Section 2.2's formulas. We only report the first and the last of the five code chunks required to carry out this work.

```

> for ii from 1 to 4*Dims do XX||ii:=t^ii-expand(subs([mu||4 = kappa1+3,
  mu||3 = Gamma1, mu||2 = 1, mu||1 = 0, mu||0 = 1], sum(binomial(ii, j)*
  'mu||j'*sigma^j*mu^(ii-j), 'j' = 0 .. ii))): end do:

> for ii from 1 to Dims do for jj from 1 to Dims do for kk from 1 to Dims
  do for ll from 1 to Dims do Mu||ii||"."||jj||"."||kk||"."||ll:=simplify(
  subs([seq(t^(4*Dims-i)=t^(4*Dims-i)-XX|| (4*Dims-i), i=0..(4*Dims-1))],
  expand(XX||ii*XX||jj*XX||kk*XX||ll)): end do; end do; end do; end do;

> ExpGaussian:=[Gamma1=0,kappa1=0]: for ii from 5 to 4*Dims do ExpGaussian:=
  [op(ExpGaussian),mu||ii=('mod'(ii+1, 2))*doublefactorial(ii-1)]: end do:
ExpGaussian;

```

$$[GAMMA1 = 0, kappa1 = 0, mu5 = 0, mu6 = 15, mu7 = 0, mu8 = 105]$$



### 3.2 Acceleration constant $a$

The purpose of the following code chunk is to compute  $A$  involved in the derivation of the value of the acceleration constant  $a$ , see Section 2.4.

```
> A:=simplify(add(add(add(PartAs([NtoL[i]])*PartAs([NtoL[j]])*
  PartAs([NtoL[k]])*Mu||i||".||j||".||k,i=1..Dims),j=1..Dims),k=1..Dims)):
  A:=simplify(subs(EspX,A),assume=positive);
```

$$A := \Gamma_1$$

As a consequence, we can state that the value of the acceleration constant  $a$  is :

$$a = \frac{\Gamma_1}{6\sqrt{n}}.$$

### 3.3 Non-studentized statistic

We now compute the values  $k_{1,2}$ ,  $k_{3,1}$ ,  $k_{2,2}$  and  $k_{4,1}$  defined in Theorem 2.3.

```
> k12:=simplify(1/2*(add(add(PartA0([NtoL[i],NtoL[j]])*Mu||i||".||j|,i=
  1..Dim),j=1..Dim))):
> k12:=simplify(subs(EspX,k12));
```

$$k12 := 0$$

```
> k31:=k31:=simplify(add(add(add(PartA0([NtoL[i]])*PartA0([NtoL[j]])*PartA0(
  [NtoL[k]])*Mu||i||".||j||".||k,i=1..Dim),j=1..Dim),k=1..Dim)+3*add(add(
  add(add(PartA0([NtoL[i]])*PartA0([NtoL[j]])*PartA0([NtoL[k],NtoL[l]])*
  Mu||i||".||k*Mu||j||".||l,i=1..Dim),j=1..Dim),k=1..Dim),l=1..Dim)):
> k31:=simplify(subs(EspX,k31));
```

$$k31 := \Gamma_1$$

It is easy to derive results for a given distribution by replacing moments with their actual values. For instance, in the Gaussian case  $\Gamma_1 = 0$  and  $\kappa_1 = 0$ . In the following, we will report only the results for the general case.

```
> eval(k31,[Gamma1=0,kappa1=0]);
```

$$0$$

```
> k22:=simplify(add(add(add(PartA0([NtoL[i]])*PartA0([NtoL[j],NtoL[k]])*
  Mu||i||".||j||".||k,i=1..Dim),j=1..Dim),k=1..Dim)+1/2*add(add(add(add(
  PartA0([NtoL[i],NtoL[j]])*PartA0([NtoL[k],NtoL[l]])*Mu||i||".||k*
  Mu||j||".||l,i=1..Dim),j=1..Dim),k=1..Dim),l=1..Dim)+add(add(add(add(
  PartA0([NtoL[i]])*PartA0([NtoL[j],NtoL[k],NtoL[l]])*Mu||i||".||j*
  Mu||k||".||l,i=1..Dim),j=1..Dim),k=1..Dim),l=1..Dim)):
> k22:=subs(EspX,k22);
```

$$k22 := 0$$

```

> k41:=simplify(add(add(add(add(PartA0([NtoL[i]])*PartA0([NtoL[j]])*PartA0([NtoL[k]])*PartA0([NtoL[l]])*(Mu||i||".||j||".||k||".||l-3*
Mu||i||".||j*Mu||k||".||l),i=1..Dim),j=1..Dim),k=1..Dim),l=1..Dim)+12*
add(add(add(add(add(PartA0([NtoL[i]])*PartA0([NtoL[j]])*PartA0([NtoL[k]])*
PartA0([NtoL[l],NtoL[m]])*Mu||i||".||l*Mu||j||".||k||".||m,i=1..Dim),
j=1..Dim),k=1..Dim),l=1..Dim),m=1..Dim)+12*add(add(add(add(add(PartA0([NtoL[i]])*PartA0([NtoL[j]])*PartA0([NtoL[k],NtoL[l]])*PartA0([NtoL[m],
NtoL[o]])*(Mu||i||".||k*Mu||j||".||m*Mu||l||".||o),i=1..Dim),j=1..Dim),
k=1..Dim),l=1..Dim),m=1..Dim),o=1..Dim)+2/3*add(add(add(add(add(PartA0([NtoL[i]])*PartA0([NtoL[j]])*PartA0([NtoL[k]])*PartA0([NtoL[l],NtoL[m],
NtoL[o]])*(6*Mu||i||".||m*Mu||k||".||l*Mu||j||".||o),i=1..Dim),
j=1..Dim),k=1..Dim),l=1..Dim),m=1..Dim),o=1..Dim)):
> k41:=subs(EspX,k41);

```

$$k41 := \kappa_1$$

In order to compute  $p_1$  and  $p_2$ , we now evaluate Equations 2 and 3.

```

> P1:=(k12+1/6*k31*(x^2-1)):
> P1:=simplify(P1);

```

$$P1 := -1/6 \Gamma_1 (x^2 - 1)$$

```

> P2:=-x*(1/2*(k22+k12^2)+1/24*(k41+4*k12*k31)*(x^2-3)+1/72*k31^2*
(x^4-10*x^2+15)):
> P2:=collect(collect(collect(P2,x),Gamma1),kappa1);

```

$$P2 := (-1/24 x^3 + 1/8 x) \kappa_1 + (-1/72 x^5 + 5/36 x^3 - 5/24 x) \Gamma_1^2$$

```

> P11:=-P1:
> P11:=collect(collect(collect(simplify(subs(EspX,P11)),x),Gamma1),kappa1);

```

$$P11 := (1/6 x^2 - 1/6) \Gamma_1$$

```

> P21:=P1*diff(P1,x)-P2-x*P1*P1/2:
> P21:=collect(collect(collect(simplify(subs(EspX,P21)),x),Gamma1),kappa1);

```

$$(1/24 x^3 - 1/8 x) \kappa_1 + (-1/18 x^3 + 5/36 x) \Gamma_1^2$$

### 3.4 Studentized statistic

$$k_{1,2,s} = -\frac{1}{2}\Gamma_1, \quad k_{3,1,s} = -2\Gamma_1, \quad k_{2,2,s} = 3 + \frac{7}{4}\Gamma_1^2, \quad k_{4,1,s} = 6 - 2\kappa_1 + 12\Gamma_1^2.$$

$$p_{1s}(x) = \frac{1}{6} (2x^2 + 1) \Gamma_1, \quad p_{2s}(x) = \left( \frac{1}{12}x^3 - \frac{1}{4}x \right) \kappa_1 + \left( -\frac{1}{18}x^5 - \frac{1}{9}x^3 + \frac{1}{6}x \right) \Gamma_1^2 - \frac{1}{4}x^3 - \frac{3}{4}x.$$

Same value as Shao and Tu [Shao and Tu, 1996, p. 145].

We now provide the expansion of the  $t$ -distribution, *cf.* remark 2.4. See the worksheet on how we used the polynomials  $p_{1s}$  and  $p_{2s}$  and an asymptotic expansion of the standard Gaussian CDF, density and polynomials  $p_{1s}$  and  $p_{2s}$  to get the polynomials  $p_{1s,t-dist}$  and  $p_{2s,t-dist}$  of the second-order Edgeworth expansion of the  $t$ -distribution.

$$p_{1s,t-dist} = p_{1s}(x), \quad p_{2s,t-dist} = p_{2s}(x) + \frac{1}{2}x.$$

These values are the one given by Lehman and Romano in Formulas 11.73 and 11.74 of Theorem 11.4.2, [Lehman and Romano, 2005, p. 460] for the second-order Edgeworth expansion of the  $t$ -distribution.

$$p_{2s,t-dist}(x) = \left(\frac{1}{12}x^3 - \frac{1}{4}x\right) \kappa_1 + \left(-\frac{1}{18}x^5 - \frac{1}{9}x^3 + \frac{1}{6}x\right) \Gamma_1^2 - \frac{1}{4}x^3 - \frac{1}{4}x.$$

$$p_{11s}(x) = -\frac{1}{6}(2x^2 + 1) \Gamma_1, \quad p_{21s}(x) = \left(-\frac{1}{12}x^3 + \frac{1}{4}x\right) \kappa_1 + \left(\frac{5}{18}x^3 - \frac{5}{72}x\right) \Gamma_1^2 + \frac{1}{4}x^3 + \frac{3}{4}x.$$

Corrects the value of  $p_{21s}$  given in Theorem 13.5 in [Das Gupta, 2008, p. 194]. Same value as the one found by Shao and Tu in [Shao and Tu, 1996, p. 145] and Finner and Dickhaus in Finner and Dickhaus [2010].

### 3.5 Export results from Maple to R

```
> fd := fopen("A_mean.R", WRITE, TEXT);

                                fd := 2

> cg1 := CodeGeneration['Matlab'](A, resultname = 'A', output = string);

                                cg1 := "A = Gamma1;"

> fprintf(fd, cg1);

                                12

> fclose(fd);
```

## 4 Expanding the variance

All the explicit results in this section were derived from the theoretical ones stated in Section 2 using **Maple**. The steps and details of the computations are available in the file **SI\_BertrandMaumy\_Edgeworth\_variance.mw**. To demonstrate how is to simple to switch from the mean expansion in Section 3 setting to another one, namely the variance expansion one, the only code chunk from that must be modified is the following one

```
> g:=proc(x1,x2) return x2-x1^2; end proc;

                                g := proc(x1,x2) return x2 - x1^2 end proc;
```

### 4.1 Acceleration constant $a$

$$A = -\frac{-\mu_6 + 3\kappa_1 + 7}{(\kappa_1 + 2)^{3/2}}.$$

In the Gaussian case, we find the same value as Shao and Tu in [Shao and Tu, 1996, p. 138] by substituting  $(\kappa_1, \mu_6)$  with the well known values  $(\kappa_1 = 0, \mu_6 = 15)$ .

$$a = \frac{1}{6\sqrt{n}} \frac{(15 - 3 \cdot 0 - 7)}{(0 + 2)^{3/2}} = \frac{\sqrt{2}}{3\sqrt{n}}.$$

## 4.2 Non-studentized statistic

$$k_{1,2} = -1/\sqrt{\kappa_1 + 2}, \quad k_{3,1} = -(-\mu_6 + 3\kappa_1 + 7 + 6\Gamma_1^2)/(\kappa_1 + 2)^{3/2}.$$

Same as the value provided by Hall in [Hall, 1992, p. 75 and 76].

$$k_{2,2} = -2(\kappa_1 + 1)/(\kappa_1 + 2), \quad k_{4,1} = (3 - 24\Gamma_1\mu_5 - 4\mu_6 + \mu_8 - 3\kappa_1^2 + 96\Gamma_1^2 - 6\kappa_1)/(\kappa_1 + 2)^2.$$

The values of  $p_1$ ,  $p_2$ ,  $p_{11}$  and  $p_{21}$  can be found in the dedicated **Maple** worksheet.

## 4.3 Studentized statistic

$$k_{1,2,s} = \frac{1}{2}(\kappa_1 + 3 - \mu_6 + 4\Gamma_1^2)/(\kappa_1 + 2)^{3/2}, \quad k_{3,1,s} = 2(-\mu_6 + 3\kappa_1 + 3\Gamma_1^2 + 7)/(\kappa_1 + 2)^{3/2}$$

Same value as the one found by Hall in [Hall, 1992, p. 76].

$$\begin{aligned} k_{2,2,s} = & \frac{1}{4}(20\kappa_1^3 + 163\kappa_1^2 + 56\Gamma_1^2\kappa_1 + 32\Gamma_1\kappa_1\mu_5 - 38\mu_6\kappa_1 + 450\kappa_1 - 90\mu_6 + 7\mu_6^2 \\ & + 415 + 112\Gamma_1^4 + 168\Gamma_1^2 + 64\Gamma_1\mu_5 - 56\Gamma_1^2\mu_6)/(\kappa_1 + 2)^3. \end{aligned}$$

$$\begin{aligned} k_{4,1,s} = & 2(6\kappa_1^3 + 84\kappa_1^2 + 297\kappa_1 + 24\Gamma_1\kappa_1\mu_5 - 32\mu_6\kappa_1 + 54\Gamma_1^2\kappa_1 - \kappa_1\mu_8 - \\ & 2\mu_8 + 312 + 72\Gamma_1^4 - 42\Gamma_1^2\mu_6 + 6\mu_6^2 + 48\Gamma_1\mu_5 + 150\Gamma_1^2 - 76\mu_6)/(\kappa_1 + 2)^3. \end{aligned}$$

The values of  $p_{1,s}$ ,  $p_{2,s}$ ,  $p_{11,s}$  and  $p_{21,s}$  can be found in the dedicated **Maple** worksheet.

# 5 A more complex example

All the explicit results in this section were derived from the theoretical ones stated in Section 2 using **Maple**. The steps and details of the computations are available in the file **SI\_BertrandMaumy\_Edgeworth\_ML\_normal\_std\_sym.mw**.

## 5.1 Introduction

One burning issue that commonly arises in official quality laboratories or in the industries of various sectors, namely chemistry, pharmacy, food processing, etc, is the estimation of the proportion of experimental results that lie within two limits. For instance, in official quality laboratories,  $[L, U]$  is the the range of the acceptable values and  $\pi$  is the proportion of valid units.

Let  $\{X_i : 1 \leq i \leq n\}$ , where  $n \geq 2$ , be a sequence of independent experimental results that follow a Gaussian distribution with mean  $\mu$  and variance  $\sigma^2 > 0$ . We denote the sample mean and the sample variance respectively by:

$$\bar{X} = \frac{1}{n} \sum_{i=1}^n X_i \quad s_c^2 = \frac{1}{n-1} \sum_{i=1}^n (X_i - \bar{X})^2.$$

We introduce the proportion  $\pi$  defined by:

$$\pi := \pi(\mu, \sigma^2) = \mathbb{P}[L \leq X \leq U] = \Phi\left(\frac{U - \mu}{\sigma}\right) - \Phi\left(\frac{L - \mu}{\sigma}\right), \quad (7)$$

where  $\Phi$  stands for the cumulative distribution function of the standard Gaussian distribution,  $L$  and  $U$ , with  $L < U$ , are the “limits of acceptance”, whose values depend on the active regulatory norm.

Using Equation 7, one can put forward  $\hat{\pi}$  as the maximum likelihood (ML) point estimator of  $\pi$ :

$$\hat{\pi} := \pi \left( \bar{X}, \frac{s_c^2}{w^2} \right) = \Phi \left( w \frac{U - \bar{X}}{s_c} \right) - \Phi \left( w \frac{L - \bar{X}}{s_c} \right), \quad (8)$$

where  $w^2 = n/(n-1)$ .

## 5.2 Towards the expansion

Following Section 2.2, let's introduce the function non-studentized function  $A_{ML,0}$  and the studentized function  $A_{ML,s}$  defined by:

$$A_{ML,0} : (x, y) \in \mathbb{R}^2 \mapsto (g_{ML}(x, y) - g_{ML}(\mu, \sigma^2 + \mu^2)) / h(\boldsymbol{\mu}) \in \mathbb{R} \quad (9)$$

$$A_{ML,s} : (x, y, z, t) \in \mathbb{R}^4 \mapsto (g_{ML}(x, y) - g_{ML}(\mu, \sigma^2 + \mu^2)) / h(x, y, z, t) \in \mathbb{R} \quad (10)$$

where  $g_{ML}(x, y) = \pi(x, y - x^2)$ ,  $\pi$  was defined in Equation 7,  $\boldsymbol{\mu} = (\mu, \mu^2 + \sigma^2, \mu^3 + 3\mu\sigma^2, \mu^4 + 6\mu^2\sigma^2 + 3\sigma^4)$  and, with  $\psi(k)$  the  $k$ th element of the vector  $\mathbf{x} = (x, y, z, t)$ , we have:

$$h_{ML}^2(\mathbf{x}) = \sum_{i=1}^2 \sum_{j=1}^2 \frac{\partial g_{ML}}{\partial \mathbf{x}_i}(x, y) \times \frac{\partial g_{ML}}{\partial \mathbf{x}_j}(x, y) \times (\psi(i+j) - \psi(i)\psi(j)).$$

Please note that  $g_{ML}(\mu, \sigma^2 + \mu^2) = \pi(\mu, \sigma^2)$ .

```
> gML:=proc(x1,x2); return 1/sqrt(2*Pi)*int(exp(-t^2/2),t=(-lambda-x1)/
sqrt(x2-x1^2)..(lambda-x1)/sqrt(x2-x1^2)); end proc;
```

```
gML := proc(x1,x2)
    return sqrt(2 * Pi)^-1 * int(exp(-1/2 * t^2), t = (-lambda - x1) *
sqrt(x2 - x1^2)^-1..(lambda - x1) * sqrt(x2 - x1^2)^-1)
end proc;
```

**Corollary 5.1.** *We have:*

$$h_{ML}(\boldsymbol{\mu}) = \frac{1}{2\pi} \left( \left( e^{-\frac{(U-\mu)^2}{2\sigma^2}} - e^{-\frac{(\mu-L)^2}{2\sigma^2}} \right)^2 + \frac{1}{2} \left( \frac{U-\mu}{\sigma} e^{-\frac{(U-\mu)^2}{2\sigma^2}} + \frac{\mu-L}{\sigma} e^{-\frac{(\mu-L)^2}{2\sigma^2}} \right)^2 \right).$$

Thanks to a direct computation Mee [1988], one can easily check that the value of  $h_{ML}(\boldsymbol{\mu})$  is equal to the asymptotic variance of  $n^{1/2}(\hat{\pi} - \pi)$  given in corollary 5.1.

We need to adapt the definition of the moments to the Gaussian frame.

```
> mu[0]:=1: mu[1]:=0: mu[2]:= 1: mu[3]:=Gamma1: Gamma1:=0: mu[4]:=kappa1+3:
kappa1:=0: for ii from 5 to 4*Dims do mu[ii]:='mod'(ii+1,2)*
doublefactorial(ii-1); end do:
> EspX:=[]:for ii in [seq(1..nops(NtoL))] do EspX:=[op(EspX),cat(x,ii)=
expand(sum(binomial(ii, j)*mu[j]*sigma^j*mu^(ii-j), j = 0 .. ii))]:
end do:EspX;
```

$$[x1 = \mu, x2 = \mu^2 + \sigma^2, x3 = \mu^3 + 3\mu\sigma^2, x4 = \mu^4 + 6\mu^2\sigma^2 + 3\sigma^4]$$

### 5.3 Expanding the estimator $\hat{\pi}$

**Theorem 5.2.** *We have the following second-order Edgeworth expansion:*

$$\begin{aligned} \mathbb{P}[\sqrt{n}F_{ML,\bullet}(\bar{X}, s_c^2/w^2) \leq x] &= \Phi(x) + n^{-1/2}p_{1,ML,\bullet}(x)\phi(x) \\ &\quad + n^{-1}p_{2,ML,\bullet}(x)\phi(x) + O(n^{-3/2}), \end{aligned} \quad (11)$$

uniformly in  $x \in \mathbb{R}$  and where  $p_{1,ML,\bullet}$  and  $p_{2,ML,\bullet}$  are defined by Equations 12 and 13 and  $\phi$  is the probability density function of a standard Gaussian distribution. Moreover, we have:

$$p_{1,ML,\bullet}(x) = -\left(k_{1,2,ML,\bullet} + \frac{1}{6}k_{3,1,ML,\bullet}(x^2 - 1)\right), \quad (12)$$

$$\begin{aligned} p_{2,ML,\bullet}(x) &= -x\left(\frac{1}{2}(k_{2,2,ML,\bullet} + k_{1,2,ML,\bullet}^2) + \frac{1}{24}(k_{4,1,ML,\bullet} + 4k_{1,2,ML,\bullet}k_{3,1,ML,\bullet})(x^2 - 3) + \right. \\ &\quad \left. \frac{1}{72}k_{3,1,ML,\bullet}^2(x^4 - 10x^2 + 15)\right). \end{aligned} \quad (13)$$

We assume, for the sake of simplifying the quantities displayed that  $\mu = 0$ . Note that  $\sigma$  remains unknown and let  $\check{\lambda} = \lambda/\sigma$ . In addition, as we stressed it before, in many regulatory issues, the limits of the interval of acceptance,  $L$  and  $U$ , are opposite numbers:  $U = \lambda = -L$ .

**Proposition 5.3** (Symmetric setting). *We begin with the values for the non-studentized statistic:*

$$\begin{aligned} \sigma_{ML}^2 &= \frac{1}{\pi}\check{\lambda}^2 e^{-\check{\lambda}^2}, \quad k_{1,2,ML,0} = \frac{1}{2\sqrt{2}}(3 - \check{\lambda}^2), \quad k_{2,2,ML,0} = \frac{3}{4}(5 - 6\check{\lambda}^2 + \check{\lambda}^4), \\ k_{3,1,ML,0} &= \frac{1}{\sqrt{2}}(5 - 3\check{\lambda}^2), \quad k_{4,1,ML,0} = 24 - 32\check{\lambda}^2 + 8\check{\lambda}^4. \end{aligned}$$

Then we follow with the values for the studentized statistic:

$$\begin{aligned} k_{1,2,ML,s} &= \frac{1}{2\sqrt{2}}(1 + \check{\lambda}^2), \quad k_{2,2,ML,s} = \frac{1}{4}(35 + 10\check{\lambda}^2 + 3\check{\lambda}^4), \quad k_{3,1,ML,s} = \frac{1}{\sqrt{2}}(-1 + 3\check{\lambda}^2), \\ k_{4,1,ML,s} &= 18 + 4\check{\lambda}^2 + 8\check{\lambda}^4. \end{aligned}$$

Hence, the second-order Edgeworth expansion of  $F_{ML,0}$  stated in Equation 11 boils down to, uniformly in  $x \in \mathbb{R}$ :

$$\begin{aligned} \mathbb{P}[\sqrt{n}F_{ML,0}(\bar{X}, s_c^2/w^2) \leq x] &= \Phi(x) + \frac{1}{\sqrt{n}}(-4 + (-5 + 3\check{\lambda}^2)x^2) \frac{\phi(x)}{6\sqrt{2}} + \frac{1}{n}\left[-x + \left(\frac{2}{3} - \right. \right. \\ &\quad \left. \left. \check{\lambda}^2 + \check{\lambda}^4\right)x^3 + \left(-\frac{25}{24} + \frac{5}{4}\check{\lambda}^2 - \frac{3}{8}\check{\lambda}^4\right)x^5\right] \frac{\phi(x)}{6} + O(n^{-3/2}). \end{aligned}$$

As to  $F_{ML,s}$ , its second-order Edgeworth expansion stated in Equation 11 boils down to, uniformly in  $x \in \mathbb{R}$ :

$$\begin{aligned} \mathbb{P}[\sqrt{n}F_{ML,s}(\bar{X}, s_c^2/w^2) \leq x] &= \Phi(x) + \frac{1}{\sqrt{n}}(-4 + (1 - 3\check{\lambda}^2)x^2) \frac{\phi(x)}{6\sqrt{2}} + \frac{1}{n}\left[-\frac{29}{2}x - \left(\frac{23}{6} + \right. \right. \\ &\quad \left. \left. 4\check{\lambda}^2 - \check{\lambda}^4\right)x^3 + \left(-\frac{1}{24} + \frac{1}{4}\check{\lambda}^2 - \frac{3}{8}\check{\lambda}^4\right)x^5\right] \frac{\phi(x)}{6} + O(n^{-3/2}). \end{aligned}$$

These results lead to Cornish-Fisher expansions of the quantiles of the estimator  $\hat{\pi}$  given by the following equation:

$$w_{\alpha}^{0,ML} = z_{\alpha} + \frac{1}{6\sqrt{2n}} (4 + (5 - 3\check{\lambda}^2)z_{\alpha}^2) + \frac{z_{\alpha}}{36n} (22 - 12\check{\lambda}^2 + (11 - 18\check{\lambda}^2 + 3\check{\lambda}^4)z_{\alpha}^2) + O\left(n^{-3/2}\right), \quad (14)$$

$$w_{\alpha}^{s,ML} = z_{\alpha} + \frac{1}{6\sqrt{2n}} (4 + (-1 + 3\check{\lambda}^2)z_{\alpha}^2) + \frac{z_{\alpha}}{36n} (79 + 12\check{\lambda}^2 + (26 + 12\check{\lambda}^2 + 3\check{\lambda}^4)z_{\alpha}^2) + O\left(n^{-3/2}\right), \quad (15)$$

where  $z_{\alpha}$  is the  $\alpha$ -level standard Normal quantile.

**Proposition 5.4** (Symmetric setting). *With the same assumptions as above, the acceleration constant  $a$  is:*

$$A = -2\sqrt{2}, \quad a = -\frac{\sqrt{2}}{3\sqrt{n}}.$$

Hence we showed that the value of the acceleration constant  $a$  does not depend of the value of either  $\sigma$  or  $\lambda$ , which could not be easily guessed from the expression of the  $\hat{\pi}$  statistic. If  $\mu \neq 0$ , unfortunately this property does no longer hold.

Nevertheless the value of the acceleration constant can be computed with **Maple** and then exported into R compatible code, which can be later sourced into an R script, using the following code chunks.

```
> fd := fopen("AML_stdsym.R", WRITE, TEXT);

                                fd := 2

> cg1 := CodeGeneration['Matlab'](AML, resultname = 'Acent', output = string);

                                cg1 := "Acent = -0.2e1 * sqrt(0.2e1);"

> cg2 := CodeGeneration['Matlab'](AMLT, resultname = 'A', output = string);
```

```

cg2 := "A = -0.2e1 * sqrt(0.2e1) * (exp(-(mu + lambda) ^ 2 / sigma ^ 2 / 0.2e1) * mu +
exp(-(mu + lambda) ^ 2 / sigma ^ 2 / 0.2e1) * lambda + exp(-(lambda - mu) ^ 2 /
sigma ^ 2 / 0.2e1) * lambda - exp(-(lambda - mu) ^ 2 / sigma ^ 2 / 0.2e1) * mu) * (0.3e1 *
exp(-(mu + lambda) ^ 2 / sigma ^ 2) * sigma ^ 2 + 0.3e1 * exp(-(lambda - mu) ^ 2
/ sigma ^ 2) * sigma ^ 2 + 0.2e1 * exp(-(mu + lambda) ^ 2 / sigma ^ 2) * mu * lambda
- 0.2e1 * exp(-(mu ^ 2 + lambda ^ 2) / sigma ^ 2) * mu ^ 2 - 0.2e1 * exp(-(lambda
- mu) ^ 2 / sigma ^ 2) * mu * lambda + 0.2e1 * exp(-(mu ^ 2 + lambda ^ 2) / sigma ^ 2)
* lambda ^ 2 + exp(-(mu + lambda) ^ 2 / sigma ^ 2) * mu ^ 2 - 0.6e1 * exp(-(mu ^ 2
+ lambda ^ 2) / sigma ^ 2) * sigma ^ 2 + exp(-(lambda - mu) ^ 2 / sigma ^ 2) * mu ^ 2
+ exp(-(mu + lambda) ^ 2 / sigma ^ 2) * lambda ^ 2 + exp(-(lambda - mu) ^ 2 /
sigma ^ 2) * lambda ^ 2) * (exp(-(mu + lambda) ^ 2 / sigma ^ 2) * mu ^ 2 - 0.2e1
* exp(-(mu ^ 2 + lambda ^ 2) / sigma ^ 2) * mu ^ 2 + exp(-(lambda - mu) ^ 2 /
sigma ^ 2) * mu ^ 2 - 0.4e1 * exp(-(mu ^ 2 + lambda ^ 2) / sigma ^ 2) * sigma ^ 2
+ 0.2e1 * exp(-(mu + lambda) ^ 2 / sigma ^ 2) * sigma ^ 2 + 0.2e1 * exp(-(lambda
- mu) ^ 2 / sigma ^ 2) * sigma ^ 2 + 0.2e1 * exp(-(mu + lambda) ^ 2 / sigma ^ 2) * mu
* lambda - 0.2e1 * exp(-(lambda - mu) ^ 2 / sigma ^ 2) * mu * lambda + 0.2e1 *
exp(-(mu ^ 2 + lambda ^ 2) / sigma ^ 2) * lambda ^ 2 + exp(-(mu + lambda) ^ 2 /
sigma ^ 2) * lambda ^ 2 + exp(-(lambda - mu) ^ 2 / sigma ^ 2) * lambda ^ 2) ^ (
-0.3e1 / 0.2e1);
"

```

```
> fprintf(fd, cg1); fprintf(fd, cg2);
```

34

1495

```
> fclose(fd);
```

We displayed on Figures 1 and 2 eight examples of approximations of the the real cumulative distribution functions (black) of the  $\hat{\pi}$  estimator. Non-increasing cumulative distribution functions resulting from the above Edgeworth expansions (see Figure 2 for blatant examples) were dealt with using rearrangement [Chernozhukov et al., 2010, Graybill et al., 2011] and plotted with dash style (- -) whereas the original cumulative distribution functions were plotted with plain style (—). In all the simulations where it is useful, rearrangement improves the approximation of the Edgeworth expansions to the true cumulative distribution functions.

In the case of the non-studentized statistics, the second-order Edgeworth expansion (blue) of the cumulative distribution function greatly improves the whole approximation to the real cumulative distribution functions over the normal approximation (red) and even over the first order Edgeworth one (green). In this non-studentized case, and even though Edgeworth expansions are asymptotically valid, the second-order one performs nicely even for very small sample sizes.

As to the studentized statistics, the second-order Edgeworth expansion (blue) of the cumulative distribution function greatly improves the approximation to the upper quantiles (the 97,5% quantile is shown by the upper pink dash-dot line) over the normal approximation (red) and even over the first order Edgeworth one (green). In this studentized case, and even though Edgeworth



expansions are asymptotically valid, the second-order one performs nicely even for small sample sizes.

Our worksheet handles the general asymmetric setting, i.e.,  $L \neq -U$  and even if, in addition,  $\mu$  or  $\sigma$  are not assigned to specific values -see the file `SI_BertrandMaumy_Edgeworth_ML_normal_gen.mw`-.

## References

- F. Bertrand and M. Maumy. Edgeworth Expansions for Two Point Estimators of the Proportion of Measures. *Comptes Rendus Mathématique*, 345:399–404, 2007.
- L.D. Brown, T.T. Cai, and A. Das Gupta. Interval Estimation for a Binomial Proportion. *Statistical Science*, 16(2):101–133, 2001.
- L.D. Brown, T.T. Cai, and A. Das Gupta. Confidence Intervals for a Binomial Proportion and Asymptotic Expansions. *The Annals of Statistics*, 30(1):160–201, 2002.
- L.D. Brown, T.T. Cai, and A. Das Gupta. Interval Estimation in Exponential Families. *Statistica Sinica*, 13:19–49, 2003.
- T.T. Cai. One-Sided Confidence Intervals in Discrete Distributions. *Journal of Statistical Planning and Inference*, 131:63–88, 2005.
- Angelo Canty and Brian Ripley. *boot: Bootstrap R (S-Plus) Functions*, 2014. URL <http://CRAN.R-project.org/package=boot>. R package version 1.3-10.
- V. Chernozhukov, I. Fernández-Val, and A. Galichon. Rearranging Edgeworth-Cornish-Fisher Expansions. *Economic Theory*, 42(2):419–435, 2010.
- A. Das Gupta. *Asymptotic Theory of Statistics and Probability*. Springer-Verlag, New York, 2008.
- A. C. Davison and D. V. Hinkley. *Bootstrap Methods and Their Applications*. Cambridge University Press, Cambridge, 1997.
- Helmut Finner and Thorsten Dickhaus. Edgeworth Expansions and Rates of Convergence for Normalized Sums: Chung’s 1946 Method Revisited. *Statistics & Probability Letters*, 80(23-24): 1875–1880, 2010. doi: 10.1016/j.spl.2010.08.014.
- W. Graybill, M. Chen, V. Chernozhukov, I. Fernandez-Val, and A. Galichon. *Rearrangement: Monotonize Point and Interval Functional Estimates by Rearrangement*, 2011. URL <http://CRAN.R-project.org/package=Rearrangement>. R package version 1.0.
- P. Hall. *The Bootstrap and Edgeworth Expansion*. Springer-Verlag, New-York, 1992.
- G.H. Hardy, J.E. Littlewood, and G. Polya. *Inequalities*. Cambridge University Press, Cambridge, 2nd edition, 1952.
- E.L. Lehman and J.P. Romano. *Testing Statistical Hypotheses*. Springer-Verlag, New York, 3rd edition, 2005.
- Maple Team. *Maple 18*. Maplesoft, a division of Waterloo Maple Inc., Waterloo, 2014.

- R.W. Mee. Estimation of the Percentage of a Normal Distribution lying outside a Specified Interval. *Communications in Statistics : Theory and Methods*, 17(5):1465–1479, 1988.
- J. Shao and D. Tu. *The Jackknife and Bootstrap*. Springer-Verlag, New York, 1996.
- R Core Team. *R: A Language and Environment for Statistical Computing*. R Foundation for Statistical Computing, Vienna, Austria, 2014. URL <http://www.R-project.org/>.
- C.S. Withers. Expansions for the Distribution and Quantiles of a Regular Functional of the Empirical Distribution with Applications to Nonparametric Confidence Intervals. *The Annals of Statistics*, 11(2):577–587, 1983.




Figure 1: Cumulative distribution functions for the non-studentized (up) and studentized (down) ML estimators for a sample size of 10 (left) and 15 (right),  $\mu = 0$ ,  $\sigma = 1$  and  $\lambda = 1$  thus  $\pi \approx 0.68$ .




Figure 2: Cumulative distribution functions for the non-studentized (up) and studentized (down) ML estimators for a sample size of 20 (left) and 30 (right),  $\mu = 0$ ,  $\sigma = 1$  and  $\lambda = 2$  thus  $\pi \approx 0.95$ .